
NAME
mawk − pattern scanning and text processing language

SYNOPSIS
mawk [−W option] [−F value] [−v var=value] [− −] ’ program text’ [file ...]
mawk [−W option] [−F value] [−v var=value] [−f program-file] [− −] [file ...]

DESCRIPTION
mawk is an interpreter for the AWK Programming Language. The AWK language is useful for manip-
ulation of data files, text retrieval and processing, and for prototyping and experimenting with algo-
rithms. mawk is anew awk meaning it implements the AWK language as defined in Aho, Kernighan
and Weinberger,The AWK Programming Language, Addison-Wesley Publishing, 1988. (Hereafter
referred to as the AWK book.)mawk conforms to the Posix 1003.2 (draft 11.3) definition of the AWK
language which contains a few features not described in the AWK book,andmawk provides a small
number of extensions.

An AWK program is a sequence ofpattern {action}pairs and function definitions. Short programs are
entered on the command line usually enclosed in ’ ’ to avoid shell interpretation. Longer programs can
be read in from a file with the −f option.Data inputis read from the list of files on the command line
or from standard input when the list is empty. The input is broken into records as determined by the
record separator variable,RS. Initially, RS = "\n" and records are synonymous with lines. Each record
is compared against eachpatternand if it matches, the program text for{action} is executed.

OPTIONS
−F value sets the field separator,FS, to value.

−f file Program text is read fromfile instead of from the command line.Multiple −f options
are allowed.

−v var=value assignsvalueto program variablevar.

− −  indicates the unambiguous end of options.

The above options will be available with any Posix compatible implementation of AWK, and imple-
mentation specific options are prefaced with−W. mawk provides six:

−W version mawk writes its version and copyright to stdout and compiled limits to stderr and
exits 0.

−W dump writesan assembler like listing of the internal representation of the program to std-
out and exits 0 (on successful compilation).

−W interactive sets unbuffered writes to stdout and line buffered reads from stdin. Records from
stdin are lines regardless of the value ofRS.

−W exec file Program text is read fromfile and this is the last option. Useful on systems that sup-
port the#! "magic number" convention for executable scripts.

−W sprintf=num adjusts the size ofmawk’s internal sprintf buffer to numbytes. Morethan rare use
of this option indicatesmawk should be recompiled.

−W posix_space forcesmawk not to consider ’\n’ to be space.

The short forms−W[vdiesp] are recognized and on some systems−We is mandatory to avoid com-
mand line length limitations.

THE AWK LANGU AGE
1. Program structure

An AWK program is a sequence ofpattern {action}pairs and user function definitions.

A pattern can be:
BEGIN
END
expression
expression , expression

One, but not both, ofpattern {action}can be omitted.If {action} is omitted it is implicitly { print }. If

patternis omitted, then it is implicitly matched.BEGIN andEND patterns require an action.

Statements are terminated by newlines, semi-colons or both. Groups of statements such as actions or
loop bodies are blocked via { ... } as in C. The last statement in a block doesn’t need a terminator.
Blank lines have no meaning; an empty statement is terminated with a semi-colon. Long statements can
be continued with a backslash, \. A statement can be broken without a backslash after a comma, left
brace, &&, ||,do, else, the right parenthesis of anif , while or for statement, and the right parenthesis of
a function definition.A comment starts with # and extends to, but does not include the end of line.

The following statements control program flow inside blocks.

if ( expr ) statement

if ( expr ) statementelsestatement

while ( expr ) statement

do statementwhile ( expr )

for ( opt_expr; opt_expr; opt_expr) statement

for ( var in array ) statement

continue

break

2. Data types, conversion and comparison
There are two basic data types, numeric and string.Numeric constants can be integer like −2, decimal
like 1.08, or in scientific notation like −1.1e4 or .28E−3.All numbers are represented internally and all
computations are done in floating point arithmetic.So for example, the expression 0.2e2 == 20 is true
and true is represented as 1.0.

String constants are enclosed in double quotes.

"This is a string with a newline at the end.\n"

Strings can be continued across a line by escaping (\) the newline. Thefollowing escape sequences are
recognized.

\\ \
\" "
\a alert,ascii 7
\b backspace,ascii 8
\t tab,ascii 9
\n newline, ascii 10
\v vertical tab, ascii 11
\f formfeed,ascii 12
\r carriagereturn, ascii 13
\ddd 1,2 or 3 octal digits for ascii ddd
\xhh 1or 2 hex digits for ascii hh

If you escape any other character \c, you get \c, i.e.,mawk ignores the escape.

There are really three basic data types; the third isnumber and stringwhich has both a numeric value
and a string value at the same time.User defined variables come into existence when first referenced
and are initialized tonull, a number and string value which has numeric value 0 and string value "".
Non-trivial number and string typed data come from input and are typically stored in fields.(See sec-
tion 4).

The type of an expression is determined by its context and automatic type conversion occurs if needed.
For example, to evaluate the statements

y = x + 2  ;  z = x  "hello"

The value stored in variable y will be typed numeric. If x is not numeric, the value read from x is con-
verted to numeric before it is added to 2 and stored in y. The value stored in variable z will be typed

string, and the value of x will be converted to string if necessary and concatenated with "hello".(Of
course, the value and type stored in x is not changed by any conversions.) Astring expression is con-
verted to numeric using its longest numeric prefix as withatof(3). A numeric expression is converted
to string by replacingexpr with sprintf(CONVFMT , expr), unlessexpr can be represented on the host
machine as an exact integer then it is converted tosprintf ("%d", expr). Sprintf() is an AWK built-in
that duplicates the functionality ofsprintf(3), andCONVFMT is a built-in variable used for internal
conversion from number to string and initialized to "%.6g". Explicit type conversions can be forced,
expr "" is string andexpr+0 is numeric.

To evaluate,expr1 rel-op expr2, if both operands are numeric or number and string then the compari-
son is numeric; if both operands are string the comparison is string; if one operand is string, the non-
string operand is converted and the comparison is string. The result is numeric, 1 or 0.

In boolean contexts such as,if ( expr ) statement, a string expression evaluates true if and only if it is
not the empty string ""; numeric values if and only if not numerically zero.

3. Regular expressions
In the AWK language, records, fields and strings are often tested for matching aregular expression.
Regular expressions are enclosed in slashes, and

expr ˜ /r/

is an AWK expression that evaluates to 1 ifexpr "matches"r , which means a substring ofexpr is in the
set of strings defined byr . With no match the expression evaluates to 0; replacing ˜ with the "not
match" operator, !˜ , rev erses the meaning.As pattern-actionpairs,

/r/ { action} and $0 ˜ /r/ { action}

are the same, and for each input record that matchesr , action is executed. Infact, /r/ is an AWK
expression that is equivalent to ($0 ˜ /r/) anywhere except when on the right side of a match operator or
passed as an argument to a built-in function that expects a regular expression argument.

AWK uses extended regular expressions as withegrep(1). Theregular expression metacharacters, i.e.,
those with special meaning in regular expressions are

ˆ $ . [ ] | ( ) * + ?

Regular expressions are built up from characters as follows:

c matches any non-metacharacterc.

\c matches a character defined by the same escape sequences used in string con-
stants or the literal characterc if \c is not an escape sequence.

. matches any character (including newline).

ˆ matches the front of a string.

$ matches the back of a string.

[c1c2c3...] matchesany character in the class c1c2c3... . An interval of characters is denoted
c1−c2 inside a class [...].

[ˆc1c2c3...] matchesany character not in the class c1c2c3...

Regular expressions are built up from other regular expressions as follows:

r1r2 matchesr1 followed immediately byr2 (concatenation).

r1 | r2 matchesr1 or r2 (alternation).

r* matchesr repeated zero or more times.

r+ matchesr repeated one or more times.

r? matchesr zero or once.

(r) matchesr, providing grouping.

The increasing precedence of operators is alternation, concatenation and unary (*, + or ?).

For example,

/ˆ[_a−zA-Z][_a−zA−Z0−9]*$/ and
/ˆ[−+]?([0−9]+\ .?|\ .[0−9])[0−9]*([eE][−+]?[0−9]+)?$/

are matched by AWK identifiers and AWK numeric constants respectively. Note that . has to be
escaped to be recognized as a decimal point, and that metacharacters are not special inside character
classes.

Any expression can be used on the right hand side of the ˜ or !˜ operators or passed to a built-in that
expects a regular expression. Ifneeded, it is converted to string, and then interpreted as a regular
expression. For example,

BEGIN { identifier = "[_a−zA−Z][_a−zA−Z0−9]*" }

$0 ˜ "ˆ" identifier

prints all lines that start with an AWK identifier.

mawk recognizes the empty regular expression, //, which matches the empty string and hence is
matched by any string at the front, back and between every character. For example,

echo abc| mawk { gsub(//, "X") ; print }
XaXbXcX

4. Records and fields
Records are read in one at a time, and stored in thefield variable $0. The record is split intofields
which are stored in$1, $2, ..., $NF. The built-in variableNF is set to the number of fields, andNR and
FNR are incremented by 1. Fields above$NF are set to "".

Assignment to$0 causes the fields andNF to be recomputed. Assignment toNF or to a field causes$0
to be reconstructed by concatenating the$i’s separated byOFS. Assignment to a field with index
greater thanNF, increasesNF and causes$0 to be reconstructed.

Data input stored in fields is string, unless the entire field has numeric form and then the type is number
and string.For example,

echo 24 24E |
mawk ’{ print($1>100, $1>"100", $2>100, $2>"100") }’
0 1 1 1

$0 and$2 are string and$1 is number and string. The first comparison is numeric, the second is string,
the third is string (100 is converted to "100"), and the last is string.

5. Expressions and operators
The expression syntax is similar to C. Primary expressions are numeric constants, string constants,
variables, fields, arrays and function calls.The identifier for a variable, array or function can be a
sequence of letters, digits and underscores, that does not start with a digit.Variables are not declared;
they exist when first referenced and are initialized tonull.

New expressions are composed with the following operators in order of increasing precedence.

assignment = += −= *= /= %= =̂

conditional ? :

logical or ||

logical and &&

array membership in
matching ˜ !˜

relational < >  <= >= == !=

concatenation (no explicit operator)

add ops + −

mul ops * / %

unary + −

logical not !

exponentiation ˆ

inc and dec ++ −− (both post and pre)

field $

Assignment, conditional and exponentiation associate right to left; the other operators associate left to
right. Any expression can be parenthesized.

6. Arrays
Awk provides one-dimensional arrays. Array elements are expressed asarray[expr]. Expr is internally
converted to string type, so, for example, A[1] and A["1"] are the same element and the actual index is
"1". Arrays indexed by strings are called associative arrays. Initiallyan array is empty; elements exist
when first accessed. An expression,expr in array evaluates to 1 ifarray[expr] exists, else to 0.

There is a form of thefor statement that loops over each index of an array.

for ( var in array ) statement

setsvar to each index of array and executesstatement. The order thatvar transverses the indices of
array is not defined.

The statement,deletearray[expr], causesarray[expr] not to exist. mawk supports an extension,delete
array, which deletes all elements ofarray.

Multidimensional arrays are synthesized with concatenation using the built-in variable SUBSEP.
array[expr1,expr2] is equivalent toarray[expr1 SUBSEPexpr2]. Testing for a multidimensional ele-
ment uses a parenthesized index, such as

if ( (i, j) in A ) print A[i, j]

7. Builtin-variables
The following variables are built-in and initialized before program execution.

ARGC number of command line arguments.

ARGV array of command line arguments, 0..ARGC-1.

CONVFMT format for internal conversion of numbers to string, initially = "%.6g".

ENVIRON array indexed by environment variables. Anenvironment string,var=value is
stored asENVIRON [var] = value.

FILENAME name of the current input file.

FNR current record number inFILENAME .

FS splits records into fields as a regular expression.

NF number of fields in the current record.

NR current record number in the total input stream.

OFMT format for printing numbers; initially = "%.6g".

OFS inserted between fields on output, initially = " ".

ORS terminates each record on output, initially = "\n".

RLENGTH length set by the last call to the built-in function,match().

RS input record separator, initially = "\n".

RSTART index set by the last call tomatch().

SUBSEP used to build multiple array subscripts, initially = "\034".

8. Built-in functions
String functions

gsub(r,s,t) gsub(r,s)
Global substitution, every match of regular expressionr in variablet is replaced by
string s. The number of replacements is returned.If t is omitted,$0 is used. An &
in the replacement strings is replaced by the matched substring oft. \& and \\ put
literal & and \, respectively, in the replacement string.

index(s,t)
If t is a substring ofs, then the position wheret starts is returned, else 0 is returned.
The first character ofs is in position 1.

length(s)
Returns the length of strings.

match(s,r)
Returns the index of the first longest match of regular expressionr in string s.
Returns 0 if no match. As a side effect, RSTART is set to the return value.
RLENGTH is set to the length of the match or −1 if no match. If the empty string is
matched,RLENGTH is set to 0, and 1 is returned if the match is at the front, and
length(s)+1 is returned if the match is at the back.

split(s,A,r) split(s,A)
Strings is split into fields by regular expressionr and the fields are loaded into array
A. The number of fields is returned. See section 11 below for more detail.If r is
omitted,FS is used.

sprintf(format,expr-list)
Returns a string constructed fromexpr-list according toformat. See the description
of printf() below.

sub(r,s,t) sub(r,s)
Single substitution, same as gsub() except at most one substitution.

substr(s,i,n) substr(s,i)
Returns the substring of strings, starting at index i , of lengthn. If n is omitted, the
suffix of s, starting ati is returned.

tolower(s)
Returns a copy of s with all upper case characters converted to lower case.

toupper(s)
Returns a copy of s with all lower case characters converted to upper case.

Arithmetic functions

atan2(y,x) Arctan ofy/x between -π andπ.

cos(x) Cosine function,x in radians.

exp(x) Exponential function.

int(x) Returnsx truncated towards zero.

log(x) Natural logarithm.

rand() Returnsa random number between zero and one.

sin(x) Sine function,x in radians.

sqrt(x) Returns square root ofx.

srand(expr) srand()
Seeds the random number generator, using the clock ifexpr is omitted, and returns
the value of the previous seed.mawk seeds the random number generator from the
clock at startup so there is no real need to call srand().Srand(expr) is useful for
repeating pseudo random sequences.

9. Input and output
There are two output statements,print andprintf .

print writes$0 ORSto standard output.

print expr1, expr2, ..., exprn
writesexpr1 OFS expr2 OFS ... exprn ORS to standard output. Numeric expressions
are converted to string withOFMT .

printf format, expr-list
duplicates the printf C library function writing to standard output.The complete
ANSI C format specifications are recognized with conversions %c, %d, %e, %E, %f,
%g, %G, %i, %o, %s, %u, %x, %X and %%, and conversion qualifiers h and l.

The argument list to print or printf can optionally be enclosed in parentheses. Print formats numbers
usingOFMT or "%d" for exact integers. "%c"with a numeric argument prints the corresponding 8 bit
character, with a string argument it prints the first character of the string.The output of print and printf
can be redirected to a file or command by appending >file, >> file or |commandto the end of the print
statement. Redirectionopensfile or commandonly once, subsequent redirections append to the already
open stream. By convention,mawk associates the filename "/dev/stderr" with stderr which allows print
and printf to be redirected to stderr. mawk also associates "−" and "/dev/stdout" with stdin and stdout
which allows these streams to be passed to functions.

The input functiongetlinehas the following variations.

getline readsinto $0, updates the fields,NF, NR andFNR.

getline <file
reads into$0 from file, updates the fields andNF.

getlinevar
reads the next record intovar, updatesNR andFNR.

getlinevar < file
reads the next record offile into var.

command| getline
pipes a record fromcommandinto $0and updates the fields andNF.

command| getlinevar
pipes a record fromcommandinto var.

Getline returns 0 on end-of-file, −1 on error, otherwise 1.

Commands on the end of pipes are executed by /bin/sh.

The functionclose(expr) closes the file or pipe associated withexpr. Close returns 0 ifexpr is an open
file, the exit status ifexpr is a piped command, and −1 otherwise.Close is used to reread a file or com-
mand, make sure the other end of an output pipe is finished or conserve file resources.

The functionfflush(expr) flushes the output file or pipe associated withexpr. Fflush returns 0 ifexpr is
an open output stream else −1.Fflush without an argument flushes stdout. Fflush with an empty argu-
ment ("") flushes all open output.

The functionsystem(expr) uses /bin/sh to executeexpr and returns the exit status of the commandexpr.
Changes made to theENVIRON array are not passed to commands executed withsystemor pipes.

10. User defined functions
The syntax for a user defined function is

function name(args) { statements}

The function body can contain a return statement

return opt_expr

A return statement is not required. Function calls may be nested or recursive. Functions are passed
expressions by value and arrays by reference. Extra arguments serve as local variables and are initial-
ized tonull. For example, csplit(s, A) puts each character ofs into arrayA and returns the length ofs.

function csplit(s, A, n, i)
{
n = length(s)
for( i = 1 ; i <= n ; i++ ) A[i] = substr(s, i, 1)
return n

}

Putting extra space between passed arguments and local variables is conventional. Functionscan be
referenced before they are defined, but the function name and the ’(’ of the arguments must touch to
avoid confusion with concatenation.

11. Splitting strings, records and files
Awk programs use the same algorithm to split strings into arrays with split(), and records into fields on
FS. mawk uses essentially the same algorithm to split files into records onRS.

Split(expr, A, sep) works as follows:

(1) If sepis omitted, it is replaced byFS. Sepcan be an expression or regular expres-
sion. If it is an expression of non-string type, it is converted to string.

(2) If sep= " " (a single space), then <SPACE> is trimmed from the front and back of
expr, and sepbecomes <SPACE>. mawk defines <SPACE> as the regular expres-
sion /[ \t\n]+/. Otherwisesep is treated as a regular expression, except that meta-
characters are ignored for a string of length 1, e.g., split(x, A, "*") and split(x, A, /\*/)
are the same.

(3) If expr is not string, it is converted to string. If expr is then the empty string "",
split() returns 0 andA is set empty. Otherwise, all non-overlapping, non-null and
longest matches ofsepin expr, separateexpr into fields which are loaded intoA. The
fields are placed in A[1], A[2], ..., A[n] and split() returns n, the number of fields
which is the number of matches plus one.Data placed inA that looks numeric is
typed number and string.

Splitting records into fields works the same except the pieces are loaded into$1, $2,..., $NF. If $0 is
empty,NF is set to 0 and all$i to "".

mawk splits files into records by the same algorithm, but with the slight difference thatRS is really a
terminator instead of a separator. (ORS is really a terminator too).

E.g., if FS = ":+" and$0 = "a::b:" , thenNF = 3 and $1 = "a", $2 = "b" and$3 = "", but if
"a::b:" is the contents of an input file andRS= ":+", then there are two records "a" and "b".

RS= " " is not special.

If FS = "", thenmawk breaks the record into individual characters, and, similarly, split(s,A,"") places
the individual characters ofs into A.

12. Multi-line records
Sincemawk interpretsRS as a regular expression, multi-line records are easy. Setting RS = "\n\n+",
makes one or more blank lines separate records.If FS = " " (the default), then single newlines, by the
rules for <SPACE> above, become space and single newlines are field separators.

For example, if a file is "ab\nc\n\n", RS = "\n\n+" andFS = " ", then there is one record
"a b\nc"with three fields "a", "b" and "c".ChangingFS = "\n", gives two fields "a b" and "c";
changingFS = "", gives one field identical to the record.

If you want lines with spaces or tabs to be considered blank, setRS = "\n([ \t]*\n)+". For compatibility
with other awks, settingRS = "" has the same effect as if blank lines are stripped from the front and
back of files and then records are determined as ifRS = "\n\n+". Posixrequires that "\n" always sepa-
rates records whenRS = "" reg ardless of the value ofFS. mawk does not support this convention,
because defining "\n" as <SPACE> makes it unnecessary.

Most of the time when you changeRS for multi-line records, you will also want to changeORS to
"\n\n" so the record spacing is preserved on output.

13. Program execution
This section describes the order of program execution. FirstARGC is set to the total number of com-
mand line arguments passed to the execution phase of the program.ARGV[0] is set the name of the

AWK interpreter andARGV[1] ... ARGV[ARGC-1] holds the remaining command line arguments
exclusive of options and program source.For example with

mawk −f prog v=1 A t=hello B

ARGC = 5 with ARGV[0] = "mawk",ARGV[1] = "v=1", ARGV[2] = "A", ARGV[3] = "t=hello" and
ARGV[4] = "B".

Next, eachBEGIN block is executed in order. If the program consists entirely ofBEGIN blocks, then
execution terminates, else an input stream is opened and execution continues.If ARGC equals 1, the
input stream is set to stdin, else the command line argumentsARGV[1] ... ARGV[ARGC-1] are
examined for a file argument.

The command line arguments divide into three sets: file arguments, assignment arguments and empty
strings "". An assignment has the formvar=string. When anARGV[i] is examined as a possible file
argument, if it is empty it is skipped; if it is an assignment argument, the assignment tovar takes place
andi skips to the next argument; elseARGV[i] is opened for input.If it fails to open, execution termi-
nates with exit code 2. If no command line argument is a file argument, then input comes from stdin.
Getline in aBEGIN action opens input. "−" as a file argument denotes stdin.

Once an input stream is open, each input record is tested against eachpattern, and if it matches, the
associatedaction is executed. Anexpression pattern matches if it is boolean true (see the end of sec-
tion 2). A BEGIN pattern matches before any input has been read, and anEND pattern matches after
all input has been read.A range pattern,expr1,expr2 , matches every record between the match of
expr1 and the matchexpr2 inclusively.

When end of file occurs on the input stream, the remaining command line arguments are examined for
a file argument, and if there is one it is opened, else theEND pattern is considered matched and all
END actionsare executed.

In the example, the assignment v=1 takes place after theBEGIN actionsare executed, and the data
placed in v is typed number and string. Input is then read from file A. On end of file A, t is set to the
string "hello", and B is opened for input. On end of file B, theEND actionsare executed.

Program flow at thepattern {action}level can be changed with the

next
exit opt_expr

statements. Anext statement causes the next input record to be read and pattern testing to restart with
the firstpattern {action}pair in the program.An exit statement causes immediate execution of the
END actions or program termination if there are none or if theexit occurs in anEND action. The
opt_exprsets the exit value of the program unless overridden by a laterexit or subsequent error.

EXAMPLES
1. emulate cat.

{ print }

2. emulate wc.

{ chars += length($0) + 1# add one for the \n
words += NF

}

END{ print NR, words, chars }

3. count the number of unique "real words".

BEGIN { FS = "[ˆA-Za-z]+" }

{ f or(i = 1 ; i <= NF ; i++) word[$i] = "" }

END { delete word[""]
for ( i in word ) cnt++
print cnt

}

4. sum the second field of every record based on the first field.

$1 ˜ /credit| gain/ { sum += $2 }
$1 ˜ /debit| loss/ {sum −= $2 }

END { print sum }

5. sort a file, comparing as string

{ l ine[NR] = $0 "" } # make sure of comparison type
# in case some lines look numeric

END { isort(line, NR)
for(i = 1 ; i <= NR ; i++) print line[i]

}

#insertion sort of A[1..n]
function isort( A, n, i, j, hold)
{
for( i = 2 ; i <= n ; i++)
{
hold = A[j = i]
while ( A[j−1] > hold )
{ j − −  ;  A[j+1] = A[j] }
A[j] = hold

}
# sentinel A[0] = "" will be created if needed

}

COMPATIBILITY ISSUES
The Posix 1003.2(draft 11.3) definition of the AWK language is AWK as described in the AWK book
with a few extensions that appeared in SystemVR4 nawk. The extensions are:

New functions: toupper() and tolower().

New variables: ENVIRON[ ] and CONVFMT.

ANSI C conversion specifications for printf() and sprintf().

New command options: −v var=value, multiple -f options and implementation options as
arguments to −W.

Posix AWK is oriented to operate on files a line at a time.RS can be changed from "\n" to another sin-
gle character, but it is hard to find any use for this — there are no examples in the AWK book. By con-
vention, RS = "", makes one or more blank lines separate records, allowing multi-line records.When
RS= "", "\n" is always a field separator regardless of the value inFS.

mawk, on the other hand, allows RS to be a regular expression. When"\n" appears in records, it is
treated as space, andFS always determines fields.

Removing the line at a time paradigm can make some programs simpler and can often improve perfor-
mance. For example, redoing example 3 from above,

BEGIN { RS = "[ˆA-Za-z]+" }

{ word[ $0 ] = "" }

END { delete word[ "" ]
for( i in word ) cnt++
print cnt

}

counts the number of unique words by making each word a record. On moderate size files,mawk
executes twice as fast, because of the simplified inner loop.

The following program replaces each comment by a single space in a C program file,

BEGIN {
RS = "/\*([ˆ*] | \*+[ˆ/*])*\*+/"

# comment is record separator
ORS = " "
getline hold

}

{ print hold ; hold = $0 }

END { printf "%s" , hold }

Buffering one record is needed to avoid terminating the last record with a space.

With mawk, the following are all equivalent,

x ˜ /a\+b/ x˜ "a\+b" x˜ "a\\+b"

The strings get scanned twice, once as string and once as regular expression. Onthe string scan,mawk
ignores the escape on non-escape characters while the AWK book advocates\c be recognized asc
which necessitates the double escaping of meta-characters in strings. Posix explicitly declines to define
the behavior which passively forces programs that must run under a variety of awks to use the more
portable but less readable, double escape.

Posix AWK does not recognize "/dev/std{out,err}" or \x hex escape sequences in strings.Unlike ANSI
C, mawk limits the number of digits that follows \x to two as the current implementation only supports
8 bit characters. The built-in fflush first appeared in a recent (1993) AT&T awk released to netlib, and
is not part of the posix standard.Aggregate deletion withdeletearray is not part of the posix standard.

Posix explicitly leaves the behavior ofFS = "" undefined, and mentions splitting the record into charac-
ters as a possible interpretation, but currently this use is not portable across implementations.

Finally, here is how mawk handles exceptional cases not discussed in the AWK book or the Posix
draft. It is unsafe to assume consistency across awks and safe to skip to the next section.

substr(s, i, n) returns the characters of s in the intersection of the closed interval [1, length(s)]
and the half-open interval [i, i+n).When this intersection is empty, the empty string is
returned; so substr("ABC", 1, 0) = "" and substr("ABC", −4, 6) = "A".

Every string, including the empty string, matches the empty string at the front so, s ˜ // and s ˜
"", are always 1 as is match(s, //) and match(s, ""). The last two setRLENGTH to 0.

index(s, t) is always the same as match(s, t1) where t1 is the same as t with metacharacters
escaped. Henceconsistency with match requires that index(s, "") always returns 1.Also the
condition, index(s,t) != 0 if and only t is a substring of s, requires index("","") = 1.

If getline encounters end of file, getline var, leaves var unchanged.Similarly, on entry to the
END actions,$0, the fields andNF have their value unaltered from the last record.

SEE ALSO
egrep(1)

Version 1.2 Dec 22 1994 11



MAWK(1) USERCOMMANDS MAWK(1)

Aho, Kernighan and Weinberger,The AWK Programming Language, Addison-Wesley Publishing,
1988, (the AWK book), defines the language, opening with a tutorial and advancing to many interesting
programs that delve into issues of software design and analysis relevant to programming in any lan-
guage.

The GAWK Manual, The Free Software Foundation, 1991, is a tutorial and language reference that does
not attempt the depth of the AWK book and assumes the reader may be a novice programmer. The sec-
tion on AWK arrays is excellent. Italso discusses Posix requirements for AWK.

BUGS
mawk cannot handle ascii NUL \0 in the source or data files.You can output NUL using printf with
%c, and any other 8 bit character is acceptable input.

mawk implements printf() and sprintf() using the C library functions, printf and sprintf, so full ANSI
compatibility requires an ANSI C library. In practice this means the h conversion qualifier may not be
available. Alsomawk inherits any bugs or limitations of the library functions.

Implementors of the AWK language have shown a consistent lack of imagination when naming their
programs.

AUTHOR
Mike Brennan (brennan@whidbey.com).

